
Vix

Blazing fast image processing for Elixir powered by libvips, the same engine that powers sharp.js.

  
    
  
  Perfect For


	Building image processing APIs and services
	Generating thumbnails at scale
	Image manipulation in web applications
	Computer vision preprocessing
	Processing large scientific/satellite images


  
    
  
  Features


	High Performance: Uses libvips' demand-driven, horizontally threaded architecture
	Memory Efficient: Processes images in chunks, perfect for large files
	Streaming Support: Read/write images without loading them fully into memory
	Rich Ecosystem: Zero-copy integration with Nx and eVision
	Zero Setup: Pre-built binaries for MacOS and Linux platforms included.
	Auto-updating API: New libvips features automatically available
	Comprehensive Documentation: Type specifications and documentation for 300+ operations


  
    
  
  Quick Start


Mix.install([
  {:vix, "~> 0.23"}
])

alias Vix.Vips.{Image, Operation}

# Create a thumbnail and optimize for web
{:ok, thumb} = Operation.thumbnail("profile.jpg", 300)
:ok = Image.write_to_file(thumb, "thumbnail.jpg", Q: 90, strip: true, interlace: true)
👉 Try in Livebook

  
    
  
  Common Operations



  
    
  
  Basic Processing


# Reading an image
{:ok, img} = Image.new_from_file("profile.jpg")

# Resize preserving aspect ratio
{:ok, resized} = Operation.resize(img, 0.5)  # 50% of original size

# Crop a section
{:ok, cropped} = Operation.crop(img, 100, 100, 500, 500)

# Rotate with white background
{:ok, rotated} = Operation.rotate(img, 90, background: [255, 255, 255])

# Smart thumbnail (preserves important features)
{:ok, thumb} = Operation.thumbnail("large.jpg", 300, size: :VIPS_SIZE_DOWN, crop: :VIPS_INTERESTING_ATTENTION)

  
    
  
  Web Optimization


# Convert to WebP with quality optimization
:ok = Image.write_to_file(img, "output.webp", Q: 80, effort: 4)

# Create progressive JPEG with metadata stripped
:ok = Image.write_to_file(img, "output.jpg", interlace: true, strip: true, Q: 85)

# Generate multiple formats
:ok = Image.write_to_file(img, "photo.avif", Q: 60)
:ok = Image.write_to_file(img, "photo.webp", Q: 80)
:ok = Image.write_to_file(img, "photo.jpg", Q: 85)

  
    
  
  Filters & Effects


# Blur
{:ok, blurred} = Operation.gaussblur(img, 3.0)

# Sharpen
{:ok, sharp} = Operation.sharpen(img, sigma: 1.0)

# Grayscale
{:ok, bw} = Operation.colourspace(img, :VIPS_INTERPRETATION_B_W)

  
    
  
  Advanced Usage


# Smart thumbnail preserving important features
{:ok, thumb} = Operation.thumbnail(
  "large.jpg",
  300,
  size: :VIPS_SIZE_DOWN, # only downsize, it will just copy if asked to upsize
  crop: :VIPS_INTERESTING_ATTENTION
)

# Process image stream on the fly
{:ok, image} =
  File.stream!("large_photo.jpg", [], 65_536)
  |> Image.new_from_enum()
# use `image` for further operations...

# Stream image to S3
:ok =
  Image.write_to_stream(image, ".png")
  |> Stream.each(&upload_chunk_to_s3/1)
  |> Stream.run()

  
    
  
  Performance


Libvips very fast and uses very little memory. See the detailed benchmark. Resizing an image is typically 4x-5x faster than using the quickest ImageMagick settings. It can also work with very large images without completely loading them to the memory.

  
    
  
  Installation


Add Vix to your dependencies:
def deps do
  [
    {:vix, "~> x.x.x"}
  ]
end
That's it! Vix includes pre-built binaries for MacOS & Linux.

  
    
  
  Advanced Setup


Want to use your system's libvips? Set before compilation:
export VIX_COMPILATION_MODE=PLATFORM_PROVIDED_LIBVIPS

See libvips installation guide for more details.

  
    
  
  Documentation & Resources


	Complete API Documentation
	Interactive Introduction (Livebook)
	Creating Rainbow Effects (Livebook)
	Auto Document Rotation (Livebook)
	Picture Language from SICP (Livebook)


  
    
  
  FAQ



  
    
  
  Should I use Vix or Image?


Image is a library which builds on top of Vix.
	Use Image when you need:
	A more Elixir-friendly API for common operations
	Higher-level operations like Blurhash
	Simple, chainable functions for common operations


	Use Vix directly when you need:
	Advanced VIPS features and fine-grained control
	Complex image processing pipelines
	Direct libvips performance and capabilities
	Lesser dependencies




  
    
  
  What image formats are supported?


Out of the box: JPEG, PNG, WEBP, TIFF, SVG, HEIF, GIF, and more. Need others? Just install libvips with the required libraries!

  
    
  
  License


MIT License - see LICENSE for details.


  

    
LICENSE
    


MIT License

Copyright (c) 2020 Akash Hiremath

Permission is hereby granted, free of charge, to any person obtaining a copy
of this software and associated documentation files (the "Software"), to deal
in the Software without restriction, including without limitation the rights
to use, copy, modify, merge, publish, distribute, sublicense, and/or sell
copies of the Software, and to permit persons to whom the Software is
furnished to do so, subject to the following conditions:

The above copyright notice and this permission notice shall be included in all
copies or substantial portions of the Software.

THE SOFTWARE IS PROVIDED "AS IS", WITHOUT WARRANTY OF ANY KIND, EXPRESS OR
IMPLIED, INCLUDING BUT NOT LIMITED TO THE WARRANTIES OF MERCHANTABILITY,
FITNESS FOR A PARTICULAR PURPOSE AND NONINFRINGEMENT. IN NO EVENT SHALL THE
AUTHORS OR COPYRIGHT HOLDERS BE LIABLE FOR ANY CLAIM, DAMAGES OR OTHER
LIABILITY, WHETHER IN AN ACTION OF CONTRACT, TORT OR OTHERWISE, ARISING FROM,
OUT OF OR IN CONNECTION WITH THE SOFTWARE OR THE USE OR OTHER DEALINGS IN THE
SOFTWARE.



  

    
Vix Introduction
    


  
    
  
  Overview


Welcome to an in-depth exploration of Vix, a powerful Elixir library for image processing. This guide will walk you through using Vix to handle common and advanced image processing tasks efficiently. Vix is particularly valuable for applications requiring high-performance image manipulation, such as web applications, scientific image analysis, digital art creation, and batch processing of large image datasets.

  
    
  
  Why Vix?


Vix provides Elixir bindings to libvips, a robust and mature image processing library. Here's why Vix stands out:
	Speed: Processes images up to 10x faster than alternatives by using sophisticated algorithms and parallel processing
	Memory efficiency: Rather than loading entire images into memory, Vix streams data in small chunks, making it ideal for processing large images
	Rich feature set: Access to hundreds of image operations from basic transformations to complex filters and analysis tools
	Elixir integration: Seamlessly integrates with Elixir's functional programming paradigm, offering immutable operations and pipeline-friendly APIs


  
    
  
  Setup


First, let's set up our development environment with the necessary dependencies:
Mix.install([
  {:vix, "~> 0.5"},
  {:kino, "~> 0.7"},  # For interactive examples
  {:req, "~> 0.4"}    # For fetching sample images
])

# Let's check our Vix version. Vix comes with pre-built libvips binaries
IO.puts("Using libvips version: " <> Vix.Vips.version())

  
    
  
  Core Concepts



  
    
  
  The Image Struct


In Vix, images are represented by %Vix.Vips.Image{} structs. Following functional programming principles, these structs are immutable - every operation creates a new image rather than modifying the existing one. This approach ensures thread safety and makes it easier to reason about image transformations.
Let's explore the various ways to create and work with images:
alias Vix.Vips.Image
alias Vix.Vips.Operation

# 1. Loading from file. Change the image path
# {:ok, from_file} = Image.new_from_file("input.jpg")

# 2. Loading from memory - useful for handling uploaded files or HTTP responses
# image_binary = File.read!("input.jpg")
# {:ok, from_binary} = Image.new_from_buffer(image_binary)

# 3. Creating a solid color image - useful for backgrounds or overlays
red = Image.build_image!(100, 100, [255, 0, 0])

# Vix implements `Kino.Render` protocol, so you can see the image by just
# returning it for the Livebook cell.
To make our examples more practical, let's create a helper module to fetch sample images and display them:
defmodule ImageHelper do
  def get_sample_image(width, height) do
    response = Req.get!("https://picsum.photos/#{width}/#{height}", decode_body: false)
    {:ok, image} = Image.new_from_buffer(response.body)
    image
  end

  def show(images, columns \\ nil)
  def show(%Image{} = img, _), do: show([img], 1)

  def show(images, columns) when is_list(images) do
    columns = columns || length(images)

    images
    |> Kino.Layout.grid(boxed: true, columns: columns)
    |> Kino.render()

    :ok
  end
end
Let's fetch a sample image to use throughout our examples:
import ImageHelper

# For our examples, let's use a sample image.
image = get_sample_image(800, 600)

  
    
  
  Image Properties and Metadata


Understanding image properties is crucial for processing. Vix provides comprehensive access to image metadata, which can be essential for making processing decisions or maintaining image information:
# Get all available metadata fields - useful for debugging and understanding image characteristics
{:ok, fields} = Image.header_field_names(image)
IO.puts("Available fields: #{Enum.join(fields, ", ")}")

# Get specific field value - useful for understanding image format and processing history
{:ok, value} = Image.header_value(image, "vips-loader")
IO.puts("Loader used: #{value}")

# Get image dimensions and number of color channels - essential for proper image manipulation
{width, height, bands} = Image.shape(image)

  
    
  
  Basic Operations



  
    
  
  Resizing and Scaling


Image resizing is one of the most common operations in image processing. Each method has its specific use case and trade-offs between speed and quality:
# Fast thumbnail generation - optimized for speed, perfect for preview generation
thumbnail = Operation.thumbnail_image!(image, 400)
# For even better performance use `Operation.thumbnail!` and pass path directly
# thumbnail = Operation.thumbnail!("input.jpg", 300)

# High-quality resize with Lanczos3 kernel - best for preserving image quality
# Lanczos3 provides excellent results for both upscaling and downscaling
resized = Operation.resize!(image, 0.5, kernel: :VIPS_KERNEL_LANCZOS3)

# Scale to specific dimensions while maintaining proper aspect ratio
# Useful for fitting images into specific containers while preventing distortion
scaled = Operation.resize!(image, 400 / Image.width(image), vscale: 300 / Image.height(image))

# Smart cropping - uses edge detection and entropy analysis to keep important parts
# Perfect for automated content-aware thumbnail generation
{smart_crop, _} = Operation.smartcrop!(image, 300, 200)

show([thumbnail, resized, scaled, smart_crop], 2)

  
    
  
  Color Operations


Color manipulation is essential for image enhancement, artistic effects, and preparing images for specific use cases:
# Convert to grayscale - useful for reducing complexity or preparing for analysis
# Also great for artistic effects or preparing images for machine learning
grayscale = Operation.colourspace!(image, :VIPS_INTERPRETATION_B_W)

# Adjust RGB channels individually for color balance
# Values > 1 increase channel intensity, < 1 decrease it
# This can correct color casts or create artistic effects
adjusted = Operation.linear!(image,
  [1.2, 1.0, 0.8],  # RGB multipliers: boost red, normal green, reduce blue
  [0, 0, 0]         # RGB offsets: no additional adjustment
)

# Extract alpha channel (transparency) if present
# Useful for masking operations or analyzing image transparency
{:ok, alpha} = Operation.extract_band(image, bands - 1)

# Add an alpha channel - useful for creating partially transparent images
# Essential for overlays and composition effects
with_alpha = Operation.bandjoin!([image, Image.build_image!(width, height, [125])])

show([grayscale, adjusted, with_alpha])

  
    
  
  Operators


Vix provides convenient operator overloading for common image manipulations. These operators make the code more readable and intuitive:
use Vix.Operator, only: [+: 2, -: 2, *: 2]  # Import specific operators

# Adjust brightness using operators - multiplication scales pixel values
brighter = image * 1.2  # Increase brightness by 20%
darker = image * 0.8    # Decrease brightness by 20%

# Combining operators for complex effects
# This creates an enhanced image with adjusted brightness, contrast, and blur
enhanced = (image * 1.2) - 10 + Operation.gaussblur!(image, 2)

show([brighter, darker, enhanced])

  
    
  
  The Access Protocol


Vix implements Elixir's Access protocol, providing a powerful way to work with image channels and regions. This makes it easy to extract and manipulate specific parts of an image:
# Get the red channel from an RGB image
# Useful for channel-specific analysis or effects
red = image[0]

# Get a 200x100 pixel square from the top-left corner
# Perfect for creating image tiles or focusing on specific regions
top_left = image[[0..199, 0..99]]

# Get the bottom-right 200x100 pixel square
# Negative indices count from the end, just like in Elixir lists
bottom_right = image[[-200..-1, -100..-1]]

# Alternatively you can use keyword list for more readable code
# Get a 200x100 pixel square from the top-left corner, and only red-green channels
red_top_left = image[[width: 0..199, height: 0..99, band: 0..1]]

show([top_left, bottom_right, red_top_left])

# Rearrange color channels - useful for color space manipulation
# RGB -> GBR: Swapping channels can create interesting color effects
remaped = Operation.bandjoin!([image[1], image[2], image[0]])

show([red, remaped])

  
    
  
  Advanced Techniques



  
    
  
  Creating Processing Pipelines


One of Vix's strengths is its ability to chain operations efficiently using Elixir's pipe operator. This allows you to create complex image processing pipelines that are both readable and performant. Each operation in the pipeline creates a new immutable image, ensuring thread safety and making it easy to debug:
defmodule ImagePipeline do
  def process(image) do
    image
    |> Operation.resize!(0.8)
    |> Operation.sharpen!()
    |> Operation.linear!([1.1], [-0.1])
  end
end

ImagePipeline.process(image)

  
    
  
  Image Composition


Combining images with text or other elements is essential for creating watermarks, annotations, or complex composite images. This example demonstrates professional text overlay with proper DPI handling and alpha channel management:
# Create high-quality text overlay with anti-aliasing
{text, _} =
  Operation.text!(
    "Hello from Vix!",
    width: 400,
    dpi: 300,         # High DPI ensures sharp text at any size
    font: "sans-serif bold",
    rgba: true        # Alpha channel enables smooth blending
  )

# Position text precisely on the image
# The embed operation handles proper padding and positioning
positioned_text =
  Operation.embed!(
    text,
    50,              # X offset from left
    50,              # Y offset from top
    Image.width(image),
    Image.height(image)
  )

# Blend using alpha composition for professional results
composite = Operation.composite2!(image, positioned_text, :VIPS_BLEND_MODE_OVER)

show([text, composite])

  
    
  
  Filters and Effects


Vix provides a comprehensive set of filters and effects suitable for both practical image enhancement and creative artistic expression. Understanding these operations allows you to create sophisticated image processing applications:
# Gaussian blur - essential for noise reduction and creating depth-of-field effects
# Higher sigma values create stronger blur effects
blurred = Operation.gaussblur!(image, 3.0)

# Edge detection using the Canny algorithm
# Perfect for image analysis and artistic effects
# Multiply by 64 to make edges more visible in the output
edges = Operation.canny!(image, sigma: 1.4) * 64

# Advanced sharpening with fine-grained control
# sigma: controls the radius of the effect
# x1: adjusts the threshold between flat and jagged areas
# m2: determines overall sharpening strength
sharpened =
  Operation.sharpen!(image,
    sigma: 1.0,
    x1: 2.0,
    m2: 20
  )

# Emboss effect using convolution matrix
# The matrix defines the relationship between each pixel and its neighbors
# This creates a 3D-like effect by emphasizing directional changes
{:ok, conv_matrix} =
  Image.new_from_list([
    [-2, -1, 0],   # Top row emphasizes vertical edges
    [-1, 1, 1],    # Middle row provides center weighting
    [0, 1, 2]      # Bottom row creates directional lighting effect
  ])

embossed = Operation.conv!(image, conv_matrix)

show([blurred, edges, sharpened, embossed], 2)

  
    
  
  Image Analysis


Vix includes powerful tools for analyzing image characteristics, making it suitable for both artistic applications and scientific image analysis. The histogram functionality is particularly useful for understanding and adjusting image exposure and color distribution:
# Create histogram from grayscale image
# Converting to grayscale first simplifies the analysis
histogram =
  image
  |> Operation.colourspace!(:VIPS_INTERPRETATION_B_W)
  |> Operation.hist_find!()

# Convert histogram to List for programmatic analysis
# Useful for automated image adjustment algorithms
Image.to_list!(histogram)
|> List.flatten()
|> IO.inspect(limit: 15, label: :histogram)

# Create visual representation of the grayscale histogram
# Normalizing ensures consistent visualization regardless of image size
bw_plot =
  histogram
  |> Operation.hist_norm!()  # Scale values to 0..255 range
  |> Operation.hist_plot!()  # Create visual representation

# Generate and plot full color histogram
# Useful for analyzing color balance and distribution
color_plot =
  image
  |> Operation.hist_find!()    # Calculate histogram for each channel
  |> Operation.hist_norm!()    # Normalize values
  |> Operation.hist_plot!()    # Create visual representation

show([bw_plot, color_plot])

  
    
  
  Error Handling


Vix offers two complementary approaches to error handling, allowing you to choose the most appropriate strategy for your application:
# Pattern 1: Explicit error handling with with
# Useful for complex workflows where you need fine-grained error control
result =
  with {:ok, resized} <- Operation.resize(image, 0.5),
       {:ok, processed} <- Operation.sharpen(resized, sigma: 1.0) do
    {:ok, processed}
  else
    {:error, reason} ->
      IO.puts("Failed: #{reason}")
      {:error, reason}
  end

# Pattern 2: Bang (!) variants for simplified error propagation
# Ideal for scripts or when you want errors to halt execution
image
|> Operation.resize!(0.5)
|> Operation.sharpen!(sigma: 1.0)

  
    
  
  Creating a Color Gradient


This example demonstrates how to programmatically generate color gradients using Vix's color space manipulation capabilities:
defmodule ColorGradient do
  def create(width, height) do
    use Vix.Operator, only: [*: 2, +: 2, /: 2]

    # Generate linear hue gradient
    # identity! creates a gradient from 0 to width
    # Multiply by 255/width normalizes values to 0..255 range
    hue = Operation.identity!(size: width, ushort: true) * 255 / width

    # Convert to HSV color space for vibrant colors
    # Add full saturation and value channels
    hsv =
      hue
      |> Operation.bandjoin_const!([255, 255])  # Add S and V channels
      |> Operation.copy!(interpretation: :VIPS_INTERPRETATION_HSV)

    # Create final gradient by repeating horizontally
    rainbow = Operation.embed!(hsv, 0, 0, width, height, extend: :VIPS_EXTEND_REPEAT)

    show([hue, hsv, rainbow], 1)
  end
end

ColorGradient.create(600, 100)

  
    
  
  Creating a Photo Collage


Create professional-looking photo collages with automatic layout, borders, and titles. This example demonstrates combining multiple images into a single composition:
defmodule Collage do
  def create(images, across, gap) do
    # create collage by arranging images in a grid pattern
    # across parameter determines number of images per row
    # gap specifies spacing between images
    collage = Operation.arrayjoin!(images, across: across, shim: gap)

    # Add a consistent border around the entire collage
    # This creates a frame effect and ensures clean edges
    {width, height, _} = Image.shape(collage)
    collage = Operation.embed!(collage, gap, gap, width + gap * 2, height + gap * 2)

    # Add a title overlay with high DPI for crisp text
    {title, _} = Operation.text!("My Photo Collage", dpi: 300)

    # Composite the title onto the collage with proper positioning
    Operation.composite2!(collage, title, :VIPS_BLEND_MODE_OVER, x: 20, y: 20)
  end
end

# Create sample collage using multiple images
# Download sample images for demonstration
images = for _ <- 1..4, do: get_sample_image(600, 400)

Collage.create(images, 2, 10)

  
    
  
  Creating Instagram-style Filters


Modern photo apps often use preset filters to enhance images. Here's how to create custom filters using Vix's operations:
defmodule PhotoFilters do
  def vintage(image) do
    image
    # Adjust color balance for warm, aged look
    |> Operation.linear!([0.9, 0.7, 0.6], [-0.1, 0.1, 0.2])
    # Add subtle blur to soften details
    |> Operation.gaussblur!(0.5)
    # Enhance contrast for dramatic effect
    |> Operation.linear!([1.2], [-0.1])
  end

  def dramatic(image) do
    # Create vignette mask using Gaussian distribution
    # This darkens the edges while keeping the center bright
    mask = Operation.gaussmat!(Image.width(image), 0.5, min: 0, max: 0.8)

    image
    # Increase contrast significantly
    |> Operation.linear!([1.4], [-0.2])
    # Enhance edge details for cinematic look
    |> Operation.sharpen!(sigma: 1.0, x1: 2.0)
    # Apply vignette effect using overlay blend
    |> Operation.composite2!(mask, :VIPS_BLEND_MODE_OVERLAY)
  end
end

# Apply our custom filters to sample image
vintage_photo = PhotoFilters.vintage(image)
dramatic_photo = PhotoFilters.dramatic(image)

show([vintage_photo, dramatic_photo])

  
    
  
  Performance Considerations


When working with Vix in production environments, keep these performance optimization strategies in mind:
	Lazy Operation Execution:
	Operations are only executed when the final result is needed
	Chain operations together to minimize intermediate processing
	Use bang variants (!) when you're confident about inputs and want to avoid error checking overhead


	Memory-Efficient Processing:
# Use sequential access for large images to reduce memory usage
{:ok, image} = Image.new_from_file("large.jpg", access: :sequential)

	Optimization Tips:
	Use thumbnail operations for quick resizes when absolute quality isn't critical
	Chain operations to avoid creating unnecessary intermediate images
	Consider format-specific loaders when you need fine-grained control
	Use appropriate color spaces for your operations (e.g., LAB for color analysis)




  
    
  
  Next Steps


To continue your journey with Vix, here are some valuable resources:
	Documentation:
	Explore the Vix documentation for comprehensive API details
	Study the libvips documentation for understanding the underlying technology


	Community:
	Join the Elixir Forum to discuss Vix with other developers
	Share your custom filters and processing pipelines with the community




  
    
  
  Advanced Topics for Further Exploration


Several advanced features deserve deeper investigation:
	Complex Convolution Operations:
	Custom kernel design for specialized filters
	Multi-pass convolutions for complex effects
	Edge handling strategies


	Color Management:
	Working with ICC profiles for color accuracy
	Color space transformations
	Calibration and profiling


	Animation Support:
	Processing animated GIFs
	Creating animations from static images
	Timeline-based effects


	Advanced Composition:
	Complex masking operations
	Layer blending modes
	Alpha channel manipulation



Remember that Vix operations are non-destructive - they always create new image objects. This makes it safe to experiment with different processing pipelines while keeping your original images intact. The functional nature of Vix operations makes it easy to compose complex transformations while maintaining code clarity and testability.


  

    
Picture Language
    


  
    
  
  Install dependencies


Mix.install([
  {:kino, "~> 0.3.0"},
  {:vix, "~> 0.5"}
])
Defining helper function display using kino so that we can show image inline
defmodule VixExt do
  alias Vix.Vips.Image
  alias Vix.Vips.Operation

  @max_height 500

  def show(%Image{} = image) do
    height = Image.height(image)

    # scale down if image height is larger than 500px
    image =
      if height > @max_height do
        Operation.resize!(image, @max_height / height)
      else
        image
      end

    # write vips-image as png image to memory
    {:ok, image_bin} = Image.write_to_buffer(image, ".png")
    Kino.render(Kino.Image.new(image_bin, "image/png"))

    :ok
  end
end

  
    
  
  Picture Language


Implementing picture language defined in Structural and Interpretation of Computer Programs section 2.2.4 in Elixir using vix
defmodule Pict do
  alias Vix.Vips.Operation, as: Op

  def beside(a, b) do
    Op.resize!(Op.join!(a, b, :VIPS_DIRECTION_HORIZONTAL), 0.5, vscale: 1)
  end

  def below(a, b) do
    Op.resize!(Op.join!(a, b, :VIPS_DIRECTION_VERTICAL), 1, vscale: 0.5)
  end

  def vert_flip(p) do
    Op.flip!(p, :VIPS_DIRECTION_VERTICAL)
  end

  def horz_flip(p) do
    Op.flip!(p, :VIPS_DIRECTION_HORIZONTAL)
  end
end
Implementation of Fig. 2.9
defmodule PictUtils do
  import Pict

  def right_split(p, 0), do: p

  def right_split(p, n) do
    t = right_split(p, n - 1)
    beside(p, below(t, t))
  end

  def up_split(p, 0), do: p

  def up_split(p, n) do
    t = up_split(p, n - 1)
    below(beside(t, t), p)
  end

  def corner_split(p, 0), do: p

  def corner_split(p, n) do
    us = up_split(p, n - 1)
    rs = right_split(p, n - 1)

    beside(
      below(beside(us, us), p),
      below(corner_split(p, n - 1), below(rs, rs))
    )
  end
end
alias Vix.Vips.Image
import VixExt

{:ok, img} = Image.new_from_file("~/Downloads/kitty.png")
img = PictUtils.corner_split(img, 5)

right = Pict.below(img, Pict.vert_flip(img))
left = Pict.horz_flip(right)
img = Pict.beside(left, right)

show(img)


  

    
Creating Rainbow  🌈
    

Mix.install([
  {:vix, "~> 0.22.0"},
  {:kino, "~> 0.10.0"}
])

  
    
  
  Introduction


Libvips provides over 300 image processing operations so getting an
intuition for what is possible, and how to combine the primitive
image processing operations, can be challenging.
In this notebook we look into some libvips core operations by
working towards a simple goal — generate a rainbow 🌈. 
The rainbow should have a half-circular arch and a smooth blend 
between the colors.

  
    
  
  Generating the colors


First, let's look into generating the rainbow colors. We need to
generate the whole spectrum with the smooth blend between them.
libvips provides the buildlut function, which generates pointwise
intermediate values between the provided points. lut here means
Look Up Table. buildlut takes a 1D matrix where each value
represents different points in the format [position, bands].
buildlut will build a single-band, one-pixel-height image with a
smooth pointwise transition between the points. If you pass [0, 0]
(at position zero pixel value is zero) [255, 255] (at position 255
value is 255), the buildlut will return an image with pixels 0 at
position 0, 1 at position 1, 2 at position 2 etc, ending with
255 at position 255. We can think of it as a range from 0 to 255, or 0..255//1.
alias Vix.Vips.Image
alias Vix.Vips.Operation

# buildlut needs matrix image.
# we can create matrix-image using `Image.new_matrix_from_array` which takes
# list and return a vips-image which can be passed to `buildlut`
{:ok, mat} = Image.new_matrix_from_array(2, 2, [[0, 0], [255, 255]])

gradient = Operation.buildlut!(mat)
Switch to attributes tab to see more details about the image.
We can see the gradient clearly if we increase the image height
Operation.resize!(gradient, 3, vscale: 50)
buildlut accepts multiple bands as well. So we can generate gradients in grayscale or color.
defmodule Vix.KinoUtils do
  # Utility to read color and parse hex string into list of
  # 8-bit integers
  def read_colors do
    start_color = Kino.Input.color("Start", default: "#DDDD55")
    end_color = Kino.Input.color("End", default: "#FF2200")

    [start_color, end_color]
    |> Kino.Layout.grid(columns: 6)
    |> Kino.render()

    start_color = read(start_color)
    end_color = read(end_color)

    {start_color, end_color}
  end

  defp read(input) do
    "#" <> color = Kino.Input.read(input)

    for <<hex::binary-size(2) <- color>> do
      String.to_integer(hex, 16)
    end
  end
end

# read user input
{start_color, end_color} = Vix.KinoUtils.read_colors()

{:ok, mat} =
  Image.new_matrix_from_array(4, 2, [
    [0 | start_color],
    [255 | end_color]
  ])

mat
|> Operation.buildlut!()
|> Operation.cast!(:VIPS_FORMAT_UCHAR)
|> Operation.resize!(3, vscale: 50)

# change the `Start` and `End` colors and evaluate
To generate the rainbow colors, the pixel values in RGB colour space
will be [255, 0, 0] (red), [255, 165, 0] (orange) ... [143, 0, 255] (violet). We could generate these values but it is inconvenient to
juggle all 3 bands.
Generating the colors in HSV color space we need [0, 255, 255] (red), [24, 255, 255] (orange) ... [212, 255, 255] (violet). Which is much
nicer to work with, since we only need to change one value.
{:ok, mat} =
  Image.new_matrix_from_array(4, 2, [
    # position 0   - [0,   255, 255]
    [0, 0, 255, 255],
    # position 255 - [255, 255, 255]
    [255, 255, 255, 255]
  ])

gradient = Operation.buildlut!(mat)

# by default the colour space won't be HSV.
# We make a shallow copy and set the colorspace to HSV.
# While also setting band format to `unsigned char`
rainbow_colors =
  Operation.copy!(gradient,
    band_format: :VIPS_FORMAT_UCHAR,
    interpretation: :VIPS_INTERPRETATION_HSV
  )

# resize to make it visible
Operation.resize!(rainbow_colors, 3, vscale: 50)

# notice that the output spectrum starts from RED and ends with RED.
# this is slightly incorrect, since rainbow ends with violet. We'll fix this later

  
    
  
  Generating the Arch


How do we generate the half circular arch? There are several way to
approach this but we are going to use complex band format and polar
coordinates.

  
    
  
  Complex Numbers


In a traditional coordinate system, a pixel’s position is represented 
by two values: its x- and y-coordinates. The x-coordinate indicates the 
horizontal position along the width of the plane, while the y-coordinate 
indicates the vertical position along the height. The two values taken 
together describe a point's location on a two-dimensional (2D) plane.
Now consider an operator, like the Fourier transformation. It takes a 
coordinate pair and may output a complex number. How would we handle that? 
It turns out that even though a complex number is a single value, it still 
consists of two components: a Real number and an Imaginary number. 
Like the x-coordinate above, the Real number component of a complex number
indicates the horizontal position along the width of the plane. An Imaginary 
number technically represents how many units, i, a point is located off a 2D plane, 
but for our purposes, we treat it the way we would a y-coordinate.
There are two different ways to locate a point on 2D plane
	Cartesian coordinate system
	Polar coordinate system

Cartesian coordinate System
Here real part represent x coordinate and imaginary part represents y coordinate.
[image: ]
a and b is used to locate the point z
Polar System
Here real part represents distance from the origin and imaginary part represents angle
[image: ]
Angle φ and distance r is used to locate point z
But why do we need complex number?
Because it makes certain operations simple. Operations which operate
on plane rather than axis. For example, to draw a circle on a polar
plane we only need to vary angle keeping the distance constant.
Let's do an example to see how libvips operations work on the complex
number plane. First, let's create an image that has an origin in the center of
the image. It makes it easy to understand what is happening.
use Vix.Operator

width = height = 255

# create 200 x 200 matrix where each pixel represent its own position.
# pixel at left-top will be `{0, 0}` (black)
# pixel right-bottom will be `{255, 255}` (white)
xy = Operation.xyz!(width, height)

# Display how axis values are chaining
Kino.Layout.grid([Kino.Text.new("X-Axis"), Kino.Text.new("Y-Axis"), xy[0], xy[1]], columns: 2)
|> Kino.render()

# move origin to center of the image
xy = (xy - [width / 2, height / 2]) * 2

# Display how axis values are changing after moving the origin.
# Notice that origin black pixel starts at top-left before
# and at center after moving the origin
Kino.Layout.grid(
  [Kino.Text.new("Centered X-Axis"), Kino.Text.new("Centered Y-Axis"), xy[0], xy[1]],
  columns: 2
)
Now that we have an image, we can see how it looks in polar plane
# convert band format to complex number format.
# we specify that read 2 bands to form a single complx band.
# x axis becomes the real part of the complex number
# y axis becomes the Imaginary part of the complex number
complex_xy = Operation.copy!(xy, format: :VIPS_FORMAT_COMPLEX, bands: 1)

# change the complex number plane to the polar plane.
# vips reads a complex number and converts it to a value on the polar plane for all pixels.
# real part will be distance from the origin
# imaginary part will be the angle in degree
polar_xy = Operation.complex!(complex_xy, :VIPS_OPERATION_COMPLEX_POLAR)

# convert the complex number back to 2-band float image.
# x axis is the real part of the complex number
# y asix is the imaginary part of the complex number
xy = Operation.copy!(polar_xy, format: :VIPS_FORMAT_FLOAT, bands: 2)

# angle will be in degree (from 0 to 360), scale it back to 255
xy = xy * [1, height / 360]

Kino.Layout.grid([Kino.Text.new("X-Axis"), Kino.Text.new("Y-Axis"), xy[0], xy[1]], columns: 2)
As you can see the x-axis is the real part of the complex number, which is
distance from the origin, and the y-axis is the imaginary part, which is the angle.
This is much easier to understand if we draw a few lines on the input
image. and see how it changes in the polar plane
defmodule ComplexOps do
  def to_polar(img, background \\ [0, 0, 0]) do
    %{width: width, height: height} = Image.headers(img)
    xy = Operation.xyz!(width, height)
    xy = xy - [width / 2, height / 2]

    scale = min(height, width) / width
    xy = xy * 2 / scale

    xy =
      xy
      |> Operation.copy!(format: :VIPS_FORMAT_COMPLEX, bands: 1)
      |> Operation.complex!(:VIPS_OPERATION_COMPLEX_POLAR)
      |> Operation.copy!(format: :VIPS_FORMAT_FLOAT, bands: 2)

    xy = xy * [1, height / 360]

    # mapim takes an input and a `map` and generates an output image
    # where input image pixels are moved based on map.
    #
    # [new_x, new_y] = map[x, y]
    # out[x, y] = img[new_x, new_y]
    #
    # mapim is to rotate, displace, distort, any type of spatial operations.
    # where the pixel value (color) remain same but the position is changed.
    Operation.mapim!(img, xy, background: background)
  end
end

x_line = Operation.black!(10, height) + 255
y_line = Operation.black!(width, 10) + 125

# create a black image and draw 2 lines
# an x axis at 50
# a y axis at 50
img =
  Operation.black!(width, height)
  |> Operation.insert!(x_line, 50, 0)
  |> Operation.insert!(y_line, 0, 50)

# convert img to polar
out = ComplexOps.to_polar(img)

Kino.Layout.grid([Kino.Text.new("Input"), Kino.Text.new("Output"), img, out], columns: 2)
A line on the x axis becomes a circle on the polar plane and a line on
the y axis becomes a line from the origin.
So to draw a rainbow circle, we just draw a rainbow line on the
x-axis and convert that to the polar plane!
rainbow_colors
|> Operation.resize!(100 / 255, vscale: 400)
|> Operation.embed!(150, 0, 600, 400)
# wrap moves the image to origin
|> Operation.wrap!()
|> ComplexOps.to_polar()
|> dbg()

# select the stage on the output to see how image transforms
:ok
All that is left now is to make a few final adjustments to make it pretty
# create colors from violet to red instead of red to red
{:ok, mat} = Image.new_matrix_from_array(4, 2, [[0, 220, 255, 255], [255, 0, 255, 255]])

rainbow_colors =
  Operation.copy!(Operation.buildlut!(mat),
    band_format: :VIPS_FORMAT_UCHAR,
    interpretation: :VIPS_INTERPRETATION_HSV
  )

sky_color = [135, 100, 255]

rainbow_colors
|> Operation.resize!(100 / 255, vscale: 500)
|> Operation.embed!(50, 0, 500, 500, background: sky_color)
|> Operation.wrap!()
|> ComplexOps.to_polar(sky_color)
# take only top half of the image
|> Operation.copy!(height: 250, width: 500)


  

    
Auto Correct Document Rotation
    

Mix.install(
  [
    {:vix, "~> 0.17.0"},
    {:kino, "~> 0.9.2"}
  ],
  # pre-built binaries does not support fourier transform operations
  # since these operations depend on an additional library.
  #
  # Usually the platform/OS provided libvips comes with these additional library
  # so we are telling vix to use the libvips provided by the platform
  # and compile NIF for that. Follow platform specific libvips
  # installation guide
  system_env: [
    {"VIX_COMPILATION_MODE", "PLATFORM_PROVIDED_LIBVIPS"}
  ]
)

  
    
  
  Introduction


In this livebook we look into correcting the text image rotation using
image processing techniques such as Fourier Transformation, complex
planes, and arithmetic operations.
This notebook is heavily based on libvips blog post
and stack overflow answer.
We use the same image mentioned on the blog to test our implementation.
So let's first fetch the test image.
alias Vix.Vips.Image
alias Vix.Vips.Operation

# import convenience math operators `+`, `-`, `*` etc.
use Vix.Operator

# we use `:httpc` to download the image
{:ok, _} = Application.ensure_all_started(:inets)
{:ok, _} = Application.ensure_all_started(:ssl)

# image link is from the stackoverflow question
image_url = 'https://i.stack.imgur.com/2q4Qr.png'
{:ok, {{_, 200, _}, _headers, bin}} = :httpc.request(:get, {image_url, []}, [timeout: 5000], [])

{:ok, img} =
  bin
  |> IO.iodata_to_binary()
  |> Image.new_from_buffer()

# convert 4 channel PNG image to black & white
img = Operation.colourspace!(img, :VIPS_INTERPRETATION_B_W)
# skip alpha band
img = img[0]
Notice that the image is not fully vertical, orienttion is slightly off

  
    
  
  Fourier Transformation


An image can be expressed as sum of sine and cosine waves of varying
magnitudes, frequency and phase. Fourier Transform is an operation
which decomposes an image into its sine and cosine components.
There are lot of resources online on this topic,
I found this
and this useful get started.
Libvips has fwfft
function for Forward Fourier Transform operation and
invfft
for Inverse Fourier Transform operation.

  
    
  
  Fwfft


fwfft returns an image with complex band format. Real part of the
band will be the wave Amplitude, Imaginary part of the band will be the
wave Phase. Position of the value is the frequency.
Since the returned image is in Complex band format, it can not be displayed.
To make it visible we need convert the complex band to 2 band float, warp the
image to center, scale values so they are visible.
white = Operation.black!(10, 200) + 255
vert_line = Operation.embed!(white, 45, 0, 200, 200)

# take fourier transform of the input image
ft = Operation.fwfft!(vert_line)

# display the images, notice the band format and band count
Kino.Layout.grid(
  [Kino.Text.new("Input"), Kino.Text.new("Fourier Transform"), vert_line, ft],
  columns: 2
)
|> Kino.render()

# convert complex number to 2 band double format
ft = Operation.copy!(ft, format: :VIPS_FORMAT_DOUBLE, bands: 2)

# do logarithm scaling for the image so that points visible
# and move the origin of the image to center
scaled_ft =
  ft
  |> Operation.scale!(log: true)
  |> Operation.wrap!()

# separate amplitude and phase channels
amp = scaled_ft[0]
phase = scaled_ft[1]

Kino.Layout.grid(
  [Kino.Text.new("Amplitude"), Kino.Text.new("Phase"), amp, phase],
  columns: 2
)
Since all these conversion is common, libvips provides
spectrum function which does all this for you.
Spectrum computes fourier transform, takes absolute value
(amplitude), scales and wraps the origin. It meant for
displaying the Fourier Transform.
Operation.spectrum!(vert_line)
Let's display fourier transform for few sample images to see how the
output changes. Change the number of lines and see how fourier transform changes.
lines_count =
  Kino.Input.number("Number of lines", default: 10)
  |> Kino.render()
  |> Kino.Input.read()

# lets create images which black and white lines
width = trunc(100 / lines_count)
black_line = Operation.black!(width, 200)

# 10 lines B&W lines
lines =
  [black_line, Operation.invert!(black_line)]
  |> List.duplicate(lines_count)
  |> List.flatten()

vert_lines = Operation.arrayjoin!(lines, across: length(lines))
horz_lines = Operation.rot!(vert_lines, :VIPS_ANGLE_D90)
vert_horz_lines = vert_lines + horz_lines

samples = [vert_lines, horz_lines, vert_horz_lines]

samples
|> Enum.flat_map(fn img ->
  [img, Operation.spectrum!(img)]
end)
|> Kino.Layout.grid(columns: 2)
As we can see, the vertical lines in the input image produces a horizontal
line in the fourier transform and horizontal lines in the
input produces to vertical line in the FT. Changing the number of lines
does not change the number lines on the output image.
So if we take Fourier Transform of a perfect text image, it should have
vertical lines and or horizontal lines exactly at 0, 90, 180, 270 degree angle,
since the characters and lines are either parallel or perpedicular. If the
document is off by some angle then the same should be visible in the Fourier
Transform.
Kino.Layout.grid([img, Operation.spectrum!(img)], columns: 2)
Indeed we can see a slightly off vertical line and horizontal
lines. Now we just need to find the angle.

  
    
  
  Finding the angle


As said before output of image of Fourier Transform will be in complex band format.
The real part of it is amplitude, which is what we are
seeing as lines and there is imaginary part which is phase.
There are two different way to plot complex numbers on a 2D plane.
	Cartesian (Rectangle) coordinate system
	Polar coordinate system

Libvips provides functions to convert numbers from one plane to other plane.
Intuitively when converting from Cartesian system to Polar system,
all vertical lines becomes the circle and horizontal lines becomes the arch/segment.
Which is what we used in the "Creating Rainbow" livebook for generating the arch.
But there is also the inverse operation. We can convert an image from Polar
plan to Cartesian plane. The circle becomes the vertical line and the
segment becomes the horizontal line. More importantly radius becomes the x-axis and
angle becomes the y-axis.
Let's see few examples
defmodule ComplexOps do
  def to_cartesian(img, background \\ [0, 0, 0]) do
    %{width: width, height: height} = Image.headers(img)
    xy = Operation.xyz!(width, height)

    # normalize the y-axis to be between 0 and 360
    xy = xy * [1, 360 / height]

    xy =
      xy
      # read values as complex numbers
      |> Operation.copy!(format: :VIPS_FORMAT_COMPLEX, bands: 1)
      # convert from polar to Cartesian plane
      |> Operation.complex!(:VIPS_OPERATION_COMPLEX_RECT)
      # and convert back to float
      |> Operation.copy!(format: :VIPS_FORMAT_FLOAT, bands: 2)

    scale = min(width, height) / width
    xy = xy * (scale / 2)
    xy = xy + [width / 2, height / 2]

    # mapim takes an input and a `map` and generates an output image
    # where input image pixels are moved based on map.
    #
    # [new_x, new_y] = map[x, y]
    # out[x, y] = img[new_x, new_y]
    #
    # mapim is to rotate, displace, distort, any type of spatial operations.
    # where the pixel value (color) remain same but the position is changed.
    Operation.mapim!(img, xy, background: background)
  end
end

samples
|> Enum.flat_map(fn img ->
  ft = Operation.spectrum!(img)
  [img, ft, ComplexOps.to_cartesian(ft)]
end)
|> Kino.Layout.grid(columns: 3)
# for the input document
img
|> Operation.spectrum!()
|> ComplexOps.to_cartesian()
Only thing left now is to find a row with maximum value.
The row number corresponding to the maximum value is the angle.
Libvips has project function which finds the row wise and column
wise sum and returns them as image, we can then use max to
find the maximum value and its position.
defmodule Utils do
  def find_angle(cartesian) do
    # find the row wise and column wise sum
    # returns 2 images with respective column/row sum
    {_columns, rows} = Operation.project!(cartesian)

    # find position of the row with maximum value
    {_, %{y: y_pos}} = Operation.max!(rows)

    # convert the y position back to angle.
    y_pos / Image.height(rows) * 360
  end
end

samples
|> Enum.flat_map(fn img ->
  ft = Operation.spectrum!(img)
  cartesian = ComplexOps.to_cartesian(ft)

  angle = Utils.find_angle(cartesian)
  # print angle next to image
  text = Kino.Text.new("\n\n\n#{to_string(angle)}")

  [img, ft, cartesian, text]
end)
|> then(fn list ->
  headers = ~w(Input Fourier-Transform Polar-Plane Angle) |> Enum.map(&Kino.Text.new/1)
  headers ++ list
end)
|> Kino.Layout.grid(columns: 4)
If there are multiple rows with same maximum values we pick one randomly.
For the input image
ft = Operation.spectrum!(img)
cartesian = ComplexOps.to_cartesian(ft)

angle = Utils.find_angle(cartesian)

# since we know that angle can only be parallel or perpendicular
# can take mod of 90
angle = angle - trunc(angle / 90) * 90

  
    
  
  Correcting the rotation


Putting it all together now we can rotate the image using the
difference as correction to fix the document
diff = 90 - angle
corrected = Operation.rotate!(img, diff)

Kino.Layout.grid([Kino.Text.new("Input"), Kino.Text.new("Corrected"), img, corrected], columns: 2)


  

    
Vix.Operator 
    



      
Provides an intuitive and readable interface for performing image processing operations by overriding common mathematical and relational operators.
This module simplifies complex image processing pipelines by allowing you to use familiar operators, such as +, -, *, /, and comparison operators (==, >=, etc.), directly with images, numbers, and lists of numbers (pixels). It also includes utility functions for logical operations and validation, like all?/2.

  
    
  
  Key Features


	Bitwise Operations: Perform Bitwise Boolean operations such as &&, ||, and xor between images and numbers.
	Arithmetic Operations: Use operators such as +, -, *, /, and ** to perform pixel-wise operations between images, numbers, or lists of numbers.
	Comparison Operations: Compare pixel values using operators like ==, !=, <, <=, >, and >=, returning the result as a new image.
	Logical Validation: Check pixel values for truthiness (e.g., 255 for true, 0 for false) using all?/2.


  
    
  
  Example Usage


defmodule Example do
 alias Vix.Vips.Operation

 def demo_operations do
   # Import only the required operators for readability and clarity
   use Vix.Operator, only: [+: 2, -: 2, *: 2, /: 2, ==: 2, all?: 2]

   # Create a black image (100x100, 3 bands)
   black = Image.build_image!(100, 100, [0, 0, 0])

   # Add constant values to pixels
   grey1 = black + 125
   grey2 = black + [125, 125, 125]

   # Pixel-wise addition: [255, 255, 255] + [0, 0, 0]
   result_image = white + grey1

   # Create a white image
   white = Image.build_image!(100, 100, [255, 255, 255])

   result = (black + 255) == white
   true = all?(result, true)
 end
end

    Vix.Tensor - vix v0.36.0
    
    

    


  
  

    
Vix.Tensor 
    



      
Struct to hold raw pixel data returned by the libvips along with metadata about the binary.
Useful for interoperability between other libraries like Nx, Evision.
See Vix.Vips.Image.write_to_tensor/1 to convert an vix image to tensor.

    Vix.Vips - vix v0.36.0
    
    

    


  
  

    
Vix.Vips 
    



      
Module for Vix.Vips.

    Vix.Vips.Image - vix v0.36.0
    
    

    


  
  

    
Vix.Vips.Image 
    



      
Primary module for reading and writing image and image metadata.
This module allows you to read, write, manipulate and analyze images efficiently using the powerful
libvips image processing library. It offers operations like loading images, accessing metadata,
and extracting image bands.

  
    
  
  Basic Usage


# Load an image from file
{:ok, image} = Image.new_from_file("path/to/image.jpg")

# Create a new RGB image
{:ok, blank} = Image.build_image(width, height, [0, 0, 0])

  
    
  
  Access Syntax (Image Slicing)


The module implements Elixir's Access behavior, providing an intuitive way to slice and extract
image data across three dimensions: width, height, and bands (color channels).

  
    
  
  Band Extraction


Access individual color bands using integer indices:
# Get the red channel from an RGB image
red_channel = image[0]

# Get the alpha channel (last band) from an RGBA image
alpha = image[-1]

  
    
  
  Band Ranges


Extract multiple consecutive bands using ranges:
# Get red and green channels from RGB
red_green = image[0..1]

# Get all channels.
all_channels = image[0..-1//1]
# Same as `all_channels = image`

  
    
  
  Dimensional Slicing


Slice images across multiple dimensions using lists of the form [width, height, bands]:
# Get a 100x100 pixel square from the top-left corner
top_left = image[[0..99, 0..99]]

# Get the bottom-right 50x50 pixel square
bottom_right = image[[-50..-1, -50..-1]]

# Get the bottom-right 50x50 pixel square, and only green channel
bottom_right = image[[-50..-1, -50..-1, 1]]

  
    
  
  Named Dimension Access


Use keyword lists for more explicit dimension specification:
# Get first 200 pixels in width, maintaining full height and bands
slice = image[[width: 0..199]]

# Get specific height range with all bands
middle = image[[height: 100..299]]

# Extract specific band
green = image[[band: 1]]

# Get a 100x100 pixel square from the top-left corner, and only red-green channels
bottom_right = image[[width: 0..99, height: 0..99, band: 0..1]]
See Vix.Vips.Operation for available image processing operations.

    Vix.Vips.Interpolate - vix v0.36.0
    
    

    


  
  

    
Vix.Vips.Interpolate 
    



      
Make interpolators for operators like affine and mapim.

    Vix.Vips.MutableImage - vix v0.36.0
    
    

    


  
  

    
Vix.Vips.MutableImage 
    



      
Vips Mutable Image
See Vix.Vips.Image.mutate/2

      


      
        Summary


  
    Types
  


    
      
        
  
    
    Vix.Vips.MutableOperation - vix v0.36.0
    
    

    


  
  

    
Vix.Vips.MutableOperation 
    



      
Module for Vix.Vips.MutableOperation.

Vix.Vips.Operation 
    



      
Provides access to VIPS operations for image processing.
This module exposes VIPS operations as Elixir functions, allowing you to perform
various image processing tasks like resizing, color manipulation, filtering,
and format conversion.

  
    
  
  Quick Start


Here's a simple example to resize an image:
# Load and resize an image to 500px width, maintaining aspect ratio
{:ok, image} = Operation.thumbnail("input.jpg", 500)

  
    
  
  Working with Operations


Operations in Vix can be grouped into several categories:
	Loading/Saving - Vix.Vips.Image, thumbnail/2, and format specific functions.
	Resizing - resize/2, thumbnail/2, smartcrop/3
	Color Management - colourspace/2, icc_transform/2
	Filters & Effects - gaussblur/2, sharpen/2
	Composition - composite/3, join/3, insert/4

Most operations follow a consistent pattern:
	Load your image
	Apply one or more operations
	Save the result


  
    
  
  Common Examples


# Basic image resizing while preserving aspect ratio
{:ok, image} = Vix.Vips.Image.new_from_file("input.jpg")
# scale down by 50%
{:ok, resized} = Operation.resize(image, scale: 0.5)
:ok = Vix.Vips.Image.write_to_file(resized, "output.jpg")

# Convert to grayscale and apply Gaussian blur
{:ok, image} = Vix.Vips.Image.new_from_file("input.jpg")
{:ok, gray} = Operation.colourspace(image, :VIPS_INTERPRETATION_B_W)
{:ok, blurred} = Operation.gaussblur(gray, 3.0)

  
    
  
  Advanced Usage



  
    
  
  Smart Cropping for Thumbnails


# Generate a smart-cropped thumbnail focusing on interesting areas
{:ok, thumb} = Operation.thumbnail("input.jpg", 300,
  crop: :attention,  # Uses image analysis to find interesting areas
  height: 300,      # Force square thumbnail
)

  
    
  
  Complex Image Composition


# Create a watermarked image with transparency
{:ok, base} = Vix.Vips.Image.new_from_file("photo.jpg")
{:ok, watermark} = Vix.Vips.Image.new_from_file("watermark.png")
{:ok, composed} = Operation.composite2(base, watermark,
  :VIPS_BLEND_MODE_OVER,  # Blend mode
  x: 20,         # Offset from left
  y: 20,         # Offset from top
  opacity: 0.8   # Watermark transparency
)

  
    
  
  Color Management


# Convert between color spaces with ICC profiles
{:ok, image} = Vix.Vips.Image.new_from_file("input.jpg")
{:ok, converted} = Operation.icc_transform(image,
  "sRGB.icc",     # Target color profile
  "input-profile": "Adobe-RGB.icc"
)

  
    
  
  Performance Tips


	Use thumbnail/2 instead of resize/2 when possible - it's optimized for common cases
	Chain operations to avoid intermediate file I/O
	For batch processing, reuse loaded ICC profiles and watermarks
	Consider using sequential mode for large images



  
    
  
  Additional Resources


	VIPS Documentation

    Vix.Vips.Image.Error - vix v0.36.0
    
    

    


  
  

    
Vix.Vips.Image.Error exception

